Всем привет.
УТ 11.2, доработка установки цен номенклатуры. Есть необходимость добавить два новых поля: Новинка и акция. У обоих тип Булево.
Добавили эти поля в табличную часть Товары.

При открытии документа формируется таблица значений следующим образом (типовой код)

```1c
    ТаблицаТоваров = Форма.Объект.Товары.Выгрузить();
```

В которой есть два новых реквизита Новинка и акция.
В таблице значений тип этих реквизитов определен как Булево, и в разных строках принимает либо Истина, либо Ложь. Т.е. значения заполнены.

Далее эту таблицу пихают в запрос в качестве параметра. (УстановкаЦенСервер.ДобавитьТоварыПредприятия())
На выходе получаем выборку с
```1c
    ОбходРезультатаЗапроса.ПоГруппировкам
```

У каждой строки значения типа булево принимают значения NULL

Почему после обработки этой таблицы запросом, булевые значения не отображаются?
---
Пытаетесь получить поля детальных записей выборки на уровне строки группировки. Спускайтесь ниже и получите значения полей.

Увидеть и понять структуру можно в отладчике, вставив код в консоль "РезультатЗапроса.Выгрузить()" вместо стандартного в коде "РезультатЗапроса.Выбрать(ОбходРезультатаЗапроса.ПоГруппировкам)"
---
Заменить

```1c
    |ИТОГИ ПО
    | Номенклатура
```

на

```1c
    |ИТОГИ
    | МАКСИМУМ(Новинка),
    | МАКСИМУМ(Акция)
    |ПО
    | Номенклатура
```
---
пихают в запрос - текст запроса ?
---
```1c
    Запрос.Текст =
    "ВЫБРАТЬ
    | Таблица.Индекс КАК Индекс,
    | Таблица.Номенклатура КАК Номенклатура,
    | Таблица.Характеристика КАК Характеристика,
    | Таблица.Новинка,
    | Таблица.Акция
    |ПОМЕСТИТЬ ТаблицаНоменклатура
    |ИЗ
    | &Таблица КАК Таблица
    |ГДЕ
    | Таблица.Номенклатура <> ЗНАЧЕНИЕ(Справочник.Номенклатура.ПустаяСсылка)
    |
    |ИНДЕКСИРОВАТЬ ПО
    | Номенклатура,
    | Характеристика
    |;
    |
    |ВЫБРАТЬ
    | ТаблицаНоменклатура.Номенклатура.Наименование КАК Наименование,
    | ТаблицаНоменклатура.Номенклатура КАК Номенклатура,
    | ТаблицаНоменклатура.Номенклатура.ЕдиницаИзмерения КАК ЕдиницаИзмерения,
    | ТаблицаНоменклатура.Номенклатура.ЦеноваяГруппа КАК ЦеноваяГруппа,
    | ТаблицаНоменклатура.Характеристика КАК Характеристика,
    | ТаблицаНоменклатура.Номенклатура.Артикул КАК Артикул,
    | ВЫБОР
    | КОГДА ТаблицаНоменклатура.Номенклатура.ИспользованиеХарактеристик В (ЗНАЧЕНИЕ(Перечисление.ВариантыИспользованияХарактеристикНоменклатуры.ОбщиеДляВидаНоменклатуры), ЗНАЧЕНИЕ(Перечисление.ВариантыИспользованияХарактеристикНоменклатуры.ОбщиеСДругимВидомНоменклатуры), ЗНАЧЕНИЕ(Перечисление.ВариантыИспользованияХарактеристикНоменклатуры.ИндивидуальныеДляНоменклатуры))
    | ТОГДА ИСТИНА
    | ИНАЧЕ ЛОЖЬ
    | КОНЕЦ КАК ХарактеристикиИспользуются,
    | ТаблицаНоменклатура.Новинка,
    | ТаблицаНоменклатура.Акция
    |ИЗ
    | ТаблицаНоменклатура КАК ТаблицаНоменклатура
    |
    |УПОРЯДОЧИТЬ ПО
    | ТаблицаНоменклатура.Индекс
    |ИТОГИ ПО
    | Номенклатура";
```
---
Пытаетесь получить поля детальных записей выборки на уровне строки группировки. Спускайтесь ниже и получите значения полей.

Увидеть и понять структуру можно в отладчике, вставив код в консоль "РезультатЗапроса.Выгрузить()" вместо стандартного в коде "РезультатЗапроса.Выбрать(ОбходРезультатаЗапроса.ПоГруппировкам)"
---
"РезультатЗапроса.Выгрузить()"
Да, я так делал. И недоумевал, почему тут есть значения а там нет.
---
Заменить

```1c
    |ИТОГИ ПО
    | Номенклатура
```

на

```1c
    |ИТОГИ
    | МАКСИМУМ(Новинка),
    | МАКСИМУМ(Акция)
    |ПО
    | Номенклатура
```
---
То что нужно. Спасибо!
---
вам помогли, отметь ответ как лучший:
 - тема пометится как закрытая
 - тем, кто столкнётся с подобной ошибкой, будет проще найти ответ.